AC_DC_GAINS

* dsauersanjose@aol.com 8/15/08
* www.idea2ic.com

*   How to Center the output voltage
*   And make AC gain externally adjustable
*   Using Only five pins.



*

```
.option set srcsteps = 1
VCC      VC       0      DC       12
VIN      IN       0      DC       0     SIN( 0 60m 10k 1n )
VOF      INP      IN     DC       .6
R1       VC       VB1    40k
R2       INN      OUT    20k
X_OPA    INN      INP    OUT      OP_AMP
R3       VAC      0      2
R4       OUT      VAC    200
CAC      INN      VAC    100u
QN1      VB1      VB1    0        NPNP
QN2      INN      VB1    0        NPNP
.tran     10n    .2m     0        10n
.control
run
set      pensize = 2
plot     out vc
.endc
.SUBCKT   OP_AMP INN    INP   OUT
EGBUF     VDIFF  0      INN   INP   1
RGM1      VDIFF  VDIFF2 1k
RGM2      VDIFF2 VGND2  1k
CBW       OUT    VGND2  4p
CSP       VDIFF2 0      4f
EGOUT     OUT    0      VGND2 0     -100000000
.ENDS     OP_AMP
```

*



```
*
.model    NPNP   NPN( BF=210 IS=1e-17   RC= 100)

.end


======AC_DC_GAINS=============
```



This circuit satisfied the automotive audio power amplifier requirement of putting out a maximum amount of power on a relatively low supply current while also allowing the audio AC gain to be externally set from 50 to 300.

A further challenge was that this was all to be done in the new TO-220 five lead power package. SGS should be credited with coming up with the clever way to used the INN pin to both center the output voltage between the supply voltage and ground while also allowing the INN pin to set the AC Gain.

The trick is that this is a power amplifier. So it resistors R3 and R4 could be make much smaller than usual. This low impedance network would be able to dominate in the AC mode over the much high DC feedback network consisting of R1, QN1, QN2, and R2.



`*#1=============WinSpiceVersion=================`

```
AC_DC_GAINS
*
*        _____ |VCC|  +12V
*      _|_     |___|
*     /VCC\      |
*     \___/      |     R1    40K                   R2  20K
*       |        |_/\  /\  /\_____ VB1 ____/\  /\  /\__
*       |            \/  \/    _|_   _|_  |  \/  \/    |
*      ///              QN1` |_| 'QN2     |             |
*                        <-'  |   |  `->  |             |
*                        _|_          _|_ |             |
*                        ///          /// |             |
*                                         |             |
*                  +.6V    |\             |             |     +6V
*      |INP|___/VOF\_______|+\            |             |  _____
*      |___|   \___/       |  \__________/ | \__________|__|OUT|
*       _|_                |  /           |                |___|
*      /VIN\               |-/            |                  |
*      \___/            ___|/             |   ____           |
*        |             |                  |  |INN|  +.6V     |
*        |             |__________________|__|___|           |
*       ///                                   _|_            |
*                                             ___  C1        |
*                                    R3   2    |             |
*                                   _/\  /\  /\_|_/\  /\  /\__|
*                                  |   \/  \/      \/  \/
*                                  |           VAC     R4  200
*                                 _|_
*                                 ///
*
.option set srcsteps = 1

VCC     VC      0      DC      12
VIN     IN      0      DC      0     SIN( 0 60m 10k 1n )
VOF     INP     IN     DC      .6
R1      VC      VB1    40k
R2      INN     OUT    20k
X_OPA   INN     INP    OUT    OP_AMP
R3      VAC     0      2
R4      OUT     VAC    200
CAC     INN     VAC    100u
QN1     VB1     VB1    0       NPNP
QN2     INN     VB1    0       NPNP
```

```
.tran    10n   .2m    0       10n

.control
run
set     pensize = 2
plot    out vc

.endc


.SUBCKT  OP_AMP INN   INP  OUT
EGBUF    VDIFF  0     INN  INP  1
RGM1     VDIFF  VDIFF2 1k
RGM2     VDIFF2 VGND2  1k
CBW      OUT    VGND2  4p
CSP      VDIFF2 0      4f
EGOUT    OUT    0     VGND2 0    -100000000
.ENDS    OP_AMP
```




```
*
.model    NPNP   NPN( BF=210 IS=1e-17   RC= 100)

.end
```